
Maple-assisted proof of recurrence for A189619

Robert Israel

20 Oct 2019

There are  configurations for a  sub-array.

 We encode these configurations as lists in the order    and enumerate them.

Configs:= [seq(convert(n,base,2)[1..8],n=2^8..2^9-1)]:
Consider the  transition matrix  with entries if the first two columns of a  sub-
array could be in configuration  while the last two columns are in configuration , and  otherwise.  
The following code computes it.
Compatible:= proc(i,j) local Xi,Xj,k;
 Xi:= Configs[i]; Xj:= Configs[j];
 if Xi[5..8] <> Xj[1..4] then return 0 fi;
 if Xi[1]=0 and ((Xi[5]=1 and Xj[5]=0) or (Xi[6]=1 and Xj[7]=0)) 
then return 0 fi;
 if Xi[2]=0 and ((Xi[6]=1 and Xj[6]=0) or (Xi[7]=1 and Xj[8]=0)) 
then return 0 fi;
 if Xi[3]=0 and ((Xi[6]=1 and Xj[5]=0) or (Xi[7]=1 and Xj[7]=0)) 
then return 0 fi;
 if Xi[4]=0 and ((Xi[7]=1 and Xj[6]=0) or (Xi[8]=1 and Xj[8]=0)) 
then return 0 fi;
 1 
end proc:
T:= Matrix(256,256,Compatible):

Thus for   where  is a column vector of all 1's.
u:= Vector(256,1):

To check, here are the first few entries of our sequence (including , which doesn't really fit the pattern,

although it will work with the recurrence).  For future use, we pre-compute more  than we need.
Tu[0]:= u:
for nn from 1 to 60 do Tu[nn]:= T . Tu[nn-1] od:
A:= [16, seq(u^%T . Tu[n],n=0..50)]:
A[1..20];

This is the empirical recurrence:
empirical:= a(n) = 8*a(n-1) -14*a(n-2) +10*a(n-3) +258*a(n-4) 

-737*a(n-5) -1502*a(n-6) +1780*a(n-7) +8033*a(n-8) -4800*a(n-9) 
-30115*a(n-10) +54016*a(n-11) -79554*a(n-12) -215482*a(n-13) 
+787373*a(n-14) -626468*a(n-15) -426934*a(n-16) +3545678*a(n-17) 
-6093224*a(n-18) +8409993*a(n-19) -6556066*a(n-20) +986540*a
(n-21) -4549030*a(n-22) +9400135*a(n-23) -23776243*a(n-24) 
+29721357*a(n-25) -36187856*a(n-26) +30281962*a(n-27) -31933555*a
(n-28) +10845312*a(n-29) +5786415*a(n-30) +7694508*a(n-31) 
+5236176*a(n-32) -9921669*a(n-33) -713908*a(n-34) +1520657*a
(n-35) -3672439*a(n-36) +1127159*a(n-37) +4759522*a(n-38) 
-673175*a(n-39) -2818276*a(n-40) -151046*a(n-41) +1043153*a(n-42)
+59137*a(n-43) -154754*a(n-44) -24424*a(n-45) +16833*a(n-46) 
+160*a(n-47) -631*a(n-48) -60*a(n-49) +4*a(n-50):

This says that  = 0 where is the following polynomial.
P:= x^50 - add(coeff(rhs(empirical),a(n-j))*x^(50-j),j=1..50);

It turns out that  Verifying this completes the proof.
R:= add(coeff(P,x,j)*Tu[j],j=0..50):
R^%T . R;

